

prácticaJP2

Práctica 2: Sucesiones, Límites y Series.

Podemos definir una función f de variable x con la que trabajar numérica o simbólicamente:

```
f(x)=x^2-3*x+1; f(1)
```

-1

```
a=var('a'); f(a); f(1+a)
```

$$\begin{aligned} &a^2 - 3a + 1 \\ &(a+1)^2 - 3a - 2 \end{aligned}$$

Recordemos que en SAGE cualquier objeto debe ser definido como perteneciente a un tipo, por eso hemos definido "a" antes de usarlo. Por otra parte veamos que tanto f como f(x) son del tipo de las expresiones simbólicas.

```
type(f); type(f(x))
```

```
< type 'sage.symbolic.expression.Expression' >
< type 'sage.symbolic.expression.Expression' >
```

No obstante son expresiones diferentes:

```
f; f(x)
```

$$\begin{aligned} &x \mapsto x^2 - 3x + 1 \\ &x^2 - 3x + 1 \end{aligned}$$

Recordemos que las sucesiones son funciones definidas sobre los naturales. Los objetivos de esta práctica son:

- Definir sucesiones reales mediante fórmulas explícitas o reglas de recurrencia y calcular límites.
- Estudiar el carácter de series.

Órdenes de SAGE

- ['expr(k)' for k in [a..b]] genera una lista con elementos expr(k) con k variando entre los enteros a y b.
- list.append(a) modifica la lista "list" añadiéndole al final el elemento "a"
- list_plot(lista) dibuja los puntos con coordenadas X: 0,1,2,... y coordenadas Y los elementos de la lista.
- limit(expr, x=x0) calcula el límite de expr cuando x tiende a x0. Se puede escribir 'lim' en vez de 'limit'.
- graf1+graf2 combina los gráficos graf1 y graf2 en uno solo.
- show(graf) muestra el gráfico graf.
- if condicion: acción si la condición es cierta, ejecuta acción.
- if condición:
 acción1
 acción2 si la condición es cierta, ejecuta varias acciones, hay que ponerlas en líneas diferentes. Es importante la indentación.
- for k in [a..b]:
 acción1
 acción2 k toma los valores desde el entero a hasta el entero b; para cada uno de dichos valores de k se ejecutan las acciones (que usualmente involucrarán a k)

2.1. Sucesiones y límites de sucesiones.

Podemos generar un número finito de términos de una sucesión. Estos elementos son dados como listas, que en SAGE son una secuencia de elementos entre corchetes y separados por comas. Por ejemplo, consideramos la sucesión

```
n=var('n')
b(n)=sin(n*pi/8)
b(10000)
```

0

Podemos generar los 10 primeros términos de la sucesión y meterlos en una lista:

```
[b(n) for n in [1..10]]
```


$$\left[\sin\left(\frac{1}{8}\pi\right), \frac{1}{2}\sqrt{2}, \sin\left(\frac{3}{8}\pi\right), 1, \sin\left(\frac{5}{8}\pi\right), \frac{1}{2}\sqrt{2}, \sin\left(\frac{7}{8}\pi\right), 0, \sin\left(\frac{9}{8}\pi\right), -\frac{1}{2}\sqrt{2} \right]$$

Así hemos calculado los 10 primeros, pero podemos determinar qué términos de la sucesión queremos que calcule:

```
[b(n) for n in [3..12]]
```

$$\left[\sin\left(\frac{3}{8}\pi\right), 1, \sin\left(\frac{5}{8}\pi\right), \frac{1}{2}\sqrt{2}, \sin\left(\frac{7}{8}\pi\right), 0, \sin\left(\frac{9}{8}\pi\right), -\frac{1}{2}\sqrt{2}, \sin\left(\frac{11}{8}\pi\right), -1 \right]$$

```
m=[b(n) for n in [0..50]]; list_plot(m)
```



Veamos algunos ejemplos de cálculo de **límites** y dibujos de sucesiones:

expr.limit(x=x0) halla el valor del límite de expr cuando x se acerca a x0

list_plot([y1,y2,...]) dibuja una lista de valores.


1) Calcular el límite de $(2n-3)^4/(3n-7)^4$; representar esta sucesión y compararla con su límite.

Estamos ante la típica indeterminación ∞/∞ .

```
n=var('n')
((-3+2*n)^4/(-7+3*n)^4).limit(n=oo)
```

$\frac{16}{81}$


```
gg1=list_plot([(-3+2*n)^4/(-7+3*n)^4 for n in [0..50]])
gg1
```



```
resta(n)=expand((-3+2*n)^4/(-7+3*n)^4-16/81).simplify_full()
resta
```

$$n \mapsto \frac{5(864n^3 - 4968n^2 + 9672n - 6371)}{81(81n^4 - 756n^3 + 2646n^2 - 4116n + 2401)}$$

```
list_plot([resta(n) for n in [0..50]])
```



2) Calcular el límite de la sucesión $a_n = \frac{(\sqrt{2n} + n)^3 - n^3}{n^2 - 2\sqrt{n^5}}$

Aquí se presenta una indeterminación del tipo $(\infty - \infty)/(\infty - \infty)$.

```
a(n)=(-n^3+(n+sqrt(2*n))^3)/(n^2-2*sqrt(n^5)); limit(a(n), n=oo)
```

$$-\frac{3}{2}\sqrt{2}$$

3) Esta no tiene límite:

```
b(n)=sin(sqrt(n+1))-sin(sqrt(n)); limit(b(n), n=oo)
```

ind

4) Calcular el límite de la sucesión $c_n = \left(\frac{3+n}{-1+n}\right)^n$

```
c(n)=(3+n)/(-1+n)^n; limit(c(n), n=oo)
```

$$e^4$$

5) Calcular el límite de la sucesión $d_n = (5n)^{1/n}$

```
d(n)=(5*n)^(1/n); limit(d(n), n=oo)
```

$$1$$

6) Calcular el límite de la sucesión $e_n = \frac{e^n}{n!}$

```
limit((e^n)/factorial(n), n=oo)
```

$$0$$

2.2. Sucesiones recurrentes


Veamos algunos ejemplos sobre sucesiones recurrentes:

1) Probar que la sucesión por $u(1)=\sqrt{2}$, $u(2)=\sqrt{2+\sqrt{2}}$, $u(3)=\sqrt{2+\sqrt{2+\sqrt{2}}}$, ... tiene límite y hallarlo.

La sucesión responde a la regla $u(n)=\sqrt{2+u(n-1)}$ con $u(1)=\sqrt{2}$.

Construiremos una lista con unos cuantos términos de la sucesión para hacernos una idea del comportamiento.

```
L=[sqrt(2)]
for n in [1..30]:
    L.append(sqrt(2+L[n-1]))
list_plot(L)
```



Esto nos sugiere que la sucesión es creciente y su límite es 2.

Pasos para estudiar la convergencia de muchas sucesiones definidas recurrentemente:

- 1) Buscar la función f asociada a la recurrencia;
- 2) Calcular los puntos fijos de f ;
- 3) Estudiar la monotonía de f ;
- 4) Estudiar la monotonía de la sucesión;
- 5) Estudiar la acotación de la sucesión;
- 6) Probar la convergencia de la sucesión y hallar el límite.

(1) Si definimos la función $f(x)=\sqrt{2+x}$, la regla es $u(n)=f(u(n-1))$.

```
f(x)=sqrt(2+x)
```

(2) Supongamos por un momento que la sucesión fuera convergente (no sabemos si lo es o no), tomando límite en la regla $u(n)=\sqrt{2+u(n-1)}$ resultaría:

$$\lim_{n \rightarrow \infty} u(n) = \sqrt{2 + \lim_{n \rightarrow \infty} u(n-1)}$$

Suponiendo que $u(n)$ es convergente, su límite L cumpliría: $L=\sqrt{2+L}$

Resolvamos entonces esta ecuación:


```
L=var('L')
solve(L==f(L),L,to_poly_solve=True)
```

```
[L = 2]
```

Esto significa que si la sucesión converge, converge a 2. Por tanto sólo resta decidir si la sucesión converge o no. Para ello:

(3) Estudiemos la monotonía de f

```
plot(f(x),-2,50)
```



En prácticas posteriores podremos estudiar la monotonía haciendo que SAGE calcule la derivada.

(4) Al ser f monótona creciente en todo su dominio, la relación de desigualdad que haya entre $u(1)$ y $u(2)$ será la misma que tendremos entre $u(n)$ y $u(n+1)$ para todo n .

Es decir, como $u(1) < u(2)$ (ya que $\sqrt{2} < \sqrt{2+\sqrt{2}}$), tomamos f a ambos lados de la desigualdad y tenemos $f(u(1)) < f(u(2))$. Es decir: $u(2) < u(3)$. Así, sucesivamente, tenemos $u(n-1) < u(n)$ para todo n . Por tanto la sucesión es monótona creciente.

(5) Ahora veamos que la sucesión es acotada. Al ser f monótona creciente en todo su dominio, la relación de desigualdad que haya entre $u(1)$ y 2 (punto fijo de f) será la misma que tendremos entre $u(n)$ y 2 para todo n . Es trivial que $u(1)=\sqrt{2} < 2$. Por tanto tomando f sucesivamente llegamos a que $f(n) < 2$ para todo n .

(6) Hemos probado que la sucesión es creciente y está acotada superiormente, así que tiene un límite real. Y, como habíamos visto al principio, en ese caso el límite es 2.

2.3. Series numéricas.

Criterios de convergencia.

Los criterios de comparación de series tienen el inconveniente de que no son intrínsecos. Es decir, es necesario encontrar otra serie con la que comparar. Por el contrario hay otros criterios que sólo dependen de la serie que estudiamos. Veamos alguno de estos criterios:

Criterio del cociente o criterio de D'Alambert

$\sum_{n=1}^{\infty} a(n)$ es convergente si $\lim_{n \rightarrow \infty} \left| \frac{a(n+1)}{a(n)} \right| < 1$

$\sum_{n=1}^{\infty} a(n)$ es divergente si $\lim_{n \rightarrow \infty} \left| \frac{a(n+1)}{a(n)} \right| > 1$

En caso de que el límite sea 1, no se puede asegurar nada.

Criterio de la raíz o criterio de Cauchy.

$\sum_{n=1}^{\infty} a(n)$ es convergente si $\lim_{n \rightarrow \infty} |a(n)|^{1/n} < 1$

$\sum_{n=1}^{\infty} a(n)$ es convergente si $\lim_{n \rightarrow \infty} |a(n)|^{1/n} > 1$

En caso de que el límite sea 1, no se puede asegurar nada.

Criterio de comparación en el límite

Sean $a(n)$ y $b(n)$ dos sucesiones tal que $\lim_{n \rightarrow \infty} \frac{a(n)}{b(n)} = L$, entonces :

i) Si $0 < L < \infty$, las series $\sum_{n=1}^{\infty} a(n)$ y $\sum_{n=1}^{\infty} b(n)$ convergen o divergen simultáneamente.

ii) Si $L=0$. Entonces, $\sum_{n=1}^{\infty} b(n)$ converge implica que $\sum_{n=1}^{\infty} a(n)$ converge.

$\sum_{n=1}^{\infty} a(n)$ diverge implica que $\sum_{n=1}^{\infty} b(n)$ diverge.

iii) Si $L=\infty$. Entonces, $\sum_{n=1}^{\infty} a(n)$ converge implica que $\sum_{n=1}^{\infty} b(n)$ converge.

$\sum_{n=1}^{\infty} b(n)$ diverge implica que $\sum_{n=1}^{\infty} a(n)$ diverge.

La serie armónica $\sum_{n=1}^{\infty} \frac{1}{n^p}$ converge si y sólo si $p > 1$.

Veamos algunos ejemplos para usar estos criterios:

1) Estudiar el carácter de la serie $\sum_{n=1}^{\infty} \frac{1}{\sqrt{n}-2/3}$

Primero notar que esta serie es de términos positivos y que su término general tiene límite 0.

```
n=var('n')
limit(1/(sqrt(n)-2/3),n=oo)
```

0

Aplicamos el criterio de comparación con la serie $\frac{1}{\sqrt{n}}$.

```
lim((1/(sqrt(n)-2/3))/(1/sqrt(n)),n=oo)
```

1

Como la serie $\sum \frac{1}{\sqrt{n}}$ diverge, se deduce que la serie $\sum \frac{1}{\sqrt{n}-2/3}$ también diverge.

2) Estudiar el carácter de la serie $\sum \frac{1+n^2}{n!}$

En este caso vamos a usar el criterio del cociente o de D'Alembert:

```
a(n)=(1+n^2)/factorial(n); lim(a(n+1)/a(n),n=oo)
```

0

Como el límite es menor que 1, la serie $\sum \frac{1+n^2}{n!}$ converge.

3) Estudiar el carácter de la serie $\sum \left(\frac{n+1}{n}\right)^{-n^3}$. Aquí usamos la raíz:

```
b(n)=((n+1)/n)^(-n^3); lim(b(n)^(1/n),n=oo)
```

0

El límite es menor que 1, luego la serie converge.

4) Estudiar el carácter de la serie $\sum \frac{n}{p^n}$.

Aplicamos el criterio del cociente:

```
n,p=var('n,p')
c(n)=n/p^n
lim(c(n+1)/c(n),n=oo)
```

$\frac{1}{p}$

Luego si $p > 1$ la serie converge, si $p < 1$ la serie diverge y si $p = 1$, obtenemos la serie de término general n , que diverge.

5) Hagamos un pequeño programa que nos analice la convergencia mediante el cociente, usando el comando "if":

```
n=var('n')
a(n)=n^n/factorial(n)
```

```
l=lim(a(n+1)/a(n),n=oo)
if l<1: print 'la serie sum',a(n),'converge'
if l>1: print 'la serie sum',a(n),'diverge'
if l==1: print 'debe aplicar otro criterio'
```

la serie sum $n^n/\text{factorial}(n)$ diverge

```
n=var('n')
a(n)=(1/n)^(n+1/n)
```

```
l=lim(a(n+1)/a(n),n=oo)
if l<1: print 'la serie sum',a(n),'converge'
if l>1: print 'la serie sum',a(n),'diverge'
if l==1: print 'debe aplicar otro criterio'
```

la serie sum $(1/n)^{(n+1/n)}$ converge

```
n=var('n')
a(n)=1/n
```

```
l=lim(a(n+1)/a(n),n=oo)
if l<1: print 'la serie sum',a(n),'converge'
if l>1: print 'la serie sum',a(n),'diverge'
if l==1: print 'debe aplicar otro criterio'
```

debe aplicar otro criterio